##*****************************************************************

## Library:     Quantum Computation Library (QCL)

## Author:      Dr. Samuel J. Lomonaco Jr.

##              (Updated by: Supreeth Hebbal)

## Description: Contains procedures and convenience functions that

##              aid in the simulation of quantum algorithms.

##*****************************************************************

with(linalg):  

with(group):
Warning, the protected names norm and trace have been redefined and unprotected

##===========================================================================

##                                   PROCEDURES

##===========================================================================
# sigma.0,sigma.1,sigma.2,sigma.3 = Predefined as the Pauli matrices

# ket.0                 = |0> = (1.0)^Transpose   

# ket.1                 = |1> = (0,1)^Transpose

# ket.i.j               = ket.i (X) ket.j   

# ket.i.j.k             = ket.i (X) ket.j (X) ket.k

# ket.i.j.k.m           = ket.i (X) ket.j (X) ket.k (X) ket.m

# Ket_To_Rho(ket)       = density operator corresponding to ket

# Make_Ket(val,dim,bs)  = Creates a Ket when given a basis state, number of qubits and 

#                         a basis as input.

# Make_Bell_Ket(val,dim)= Creates a ket in the Bell basis.

# Make Std_Ket(val,dim) = Creates a ket in the Standard basis.

# Normalize_Ket(ket k)  = Normalizes a ket using the 2-norm.

# RKet(ket)             = column vector in symbolic ket form

# Probability_Points    = Returns the probability points for a ket

# Plot_Ket(ket k)       = Creates a Maple scatterplot for a ket.              

# Measure_One_Qubit     = Performs a measurement on a single qubit in a ket.

# Measure_All_Qubits    = Performs a measurement on a all qubits in a ket.

# General_Measurement   = Performs the measurement of the density matrix.

# General_Measurment_Ket= Performs the measurement of a ket.

# Small_Transformation  = Computes a ket when we apply a small trans to a qubit register

# Swap_Qubits           = Returns the result of swapping the positions of qubit bit 1 and qubit #                         bit 2 in the original ket.

# Int_To_Bin(n)         = postive integer n written as a binary number of length lgth

# Bell.i.j              = Bell basis kets, i = 0..1, j = 0..1

# Bell.i.j.k            = Bell basis kets, i = 0..1, j = 0..1, k = 0..1

# Bell.i.j.k.m          = Bell basis kets, i = 0..1, j = 0..1, k = 0..1, m = 0..1

# Bell4(a,b,c,d)        = ( ket.(1-a).(1-b).(1-c).(1-d) + ((-1)^a)*ket.a.b.c.d)/sqrt(2)

#                         where a,b,c,d are elts of {0,1}

# Sam.i.j.k             = Sam basis kets, i = 0..1, j = 0..1, k = 0..1

# kr:=proc(A,B)         = Computes the tensor product A(X)B of matrices A and B

# QF:=proc(U)           = Computes a unitary matrix QF such that QF &* U &* is a diagonal matrix

# L3x3.j,  j=1..3       = Predefined infinitesimal generators of SO(3) 

# LL.j,    j=1..6       = Predefined infinitesimal generators of SO(4)

# FLL4:=proc(LU)        = Expresses a skew symmetric 4x4 matrix as a linear 

#                         combination of LL.i's.

# SSS.i                 = sigma.i

# SS.(i+4*j)            = kr(sigma.i,sigma.j) are predefined

# SSS.i.j=SS.i.j        = kr(sigma.i,sigma.j) are predefined

# SSS.i.j.k             = kr( SS.(i+4*j), sigma.k ) are predefined

# SSS.i.j.k.m           = kr( SS.(i+4*j), SS(k+4*m) ) are predefined

# LogU:=proc(U)         = Computes natural log of the square matrix U

# Logz(Z)               = Computes the natural log of z

# Pauli1:=proc(U)       = Expresses a 2x2 Hermitian matrix as a linear sum of sigma.i's

# RPauli1:=proc(LU)     = Expresses a 2x2 Hermitian matrix as a linear formal sum of sigma.i's

# Pauli2:=proc(LU)      = Expresses a 4x4 Hermitian matrix as a linear sumof sigma.i(X)sigma.j's

# RPauli2:=proc(rho)    = Expresses density operator rho as a formal sum of sigma.i(X)sigma.j's

# Inv_Pauli2:=proc(R)   = For a given a 4x4 matrix R over the reals, this procedure

#                         computes  rho = Sum(p=0..3, q=0..3) R[p+1,q+q]*sigma.p(X)sigma.q

# Pauli3:=proc(LU)      = Expresses a 8x8 Hermitian matrix as a linear sum of

#                         sigma.i (X) sigma.j (X) sigma.k's

# RPauli3(rho)          = Expresses density operator rho as a formal sum of 

#                         sigma.i(X)sigma.j(X)sigma.k's

# Pauli4:=proc(LU)      = Expresses a 16x16 Hermitian matrix as a linear sum of

#                         sigma.i (X) sigma.j (X) sigma.k (X) sigma.m's

# RPauli4(rho)          = Expresses density operator rho as a formal sum of 

#                         sigma.i(X)sigma.j(X)sigma.k(X)sigma.m's

# Matrep:=proc(P,P_Deg) = Converts the degree P_Deg permutation P (represented as

#                         a list of lists) into a P_Deg x P_Deg permutation matrix

# Fourier:=proc(n)      = Computes the nxn matrix (omega^ij)/sqrt(n), where omega=exp(2PiI/n)

# Vec:=proc(A)          = Transforms an mxn matrix A into an mxn column vector V, 

#                         column by column

# Ad:=proc(Q)           = Computes Ad_Q as a 16x16 matrix; Q is assumed to be a 

#                         4x4 unitary matrix

# Small_ad:=proc(Q)(M)  = Small_Ad_iQ(M), where Q is Hermitian and M is skew-Hermitian

# f0,f1,f2,f3,f4,f5     = Helper functions called by Trace1, Trace2, Trace3

# Trace1,2,3,4,5,6      = Partial trace functions

# TraceRL(A)            = Sum(a=0..1, A[a+2*i1,j0+2*a] )

# TraceLR(A)            = Sum(a=0..1, A[i0+2*a,a+2*j1] )

# PTrace(n,rho)         = Partial trace of the n-th qubit = (Trace.(n+1))(rho)  

#                         Qubits are listed right to left, and labeled 0,1, ...

#                         n = 0,1,2,3,4,5

# FCom(A,B)             = Computes the commutator [A,B] of matrices A and B

# RemoveIPi             = proc(A) = A/(I*Pi), where A is a square matrix

# My_Conj(z)            = Computes the conjugate of the comple number z

# My_Adjoint(A)         = Computes the adjoint of the complex square matrix  A

# Ket                   = (a,b)->matrix(2,1,[cos(a),(exp(I*b)*sin(a))]):

# Vec                   = proc(M), Vec(M) = Matrix( Dim,1, [col1(M),col2(M), 

#                         ... , col.CD(M)]), where Dim = rowdim(M)*coldim(M), 

#                         and CD = coldim(M)

# Id(n)                 = Computes the nxn identity matrix

# CNOT12                = (i,j)->(i,i+j)   ....  CNOT(21):=(i,j)->(i+j,j)

# Hadamard(n)           = (X)_(i=1..n) H, where H is the 2x2 Hadarmard matrix

# CNOT(T,C,W)           = permutation (written as a product of disjoint cyles) which 

#                         represents a CNOT with target bit T, control bit C, and with 

#                         W wires labeled 0..(W-1).

# norm2(X)              = Computes the square root of the spectral radius of

#                         evalm( X &* map(conjugate, transpose(X))) 

###################################################################

## Function:    sigma

## Description: Set of convenience functions that represent the 

##              Pauli spin matrices.

##              sigma||0, sigma||1, sigma||2, sigma||3 

## Inputs:      None.

## Outputs:     A Pauli spin matrix.

###################################################################

sigma||0:=diag(1,1):

sigma||1:=matrix(2,2,[0,1,1,0]):

sigma||2:=matrix(2,2,[0,-I,I,0]):

sigma||3:=matrix(2,2,[1,0,0,-1]):

###################################################################

## Function:    ket, Ket_To_Rho

## Description: Set of convenience functions that represent kets

##              as column vectors. The following kets are supported:

##              ket||0, ket||1, ket||i||j, ket||i||j||k, 

##              ket||i||j||k||m

## Inputs:      None.

## Outputs:     A column vector representing a ket.

###################################################################

## ket 0

ket||0:=matrix(2,1,[1,0]):

## ket 1

ket||1:=matrix(2,1,[0,1]):

## ket||i||j

for i from 0 to 1 do   

   for j from 0 to 1 do

      ket||i||j:=kr(ket||i,ket||j);

   od;   

od;

## ket||i||j||k

for i from 0 to 1 do   

   for j from 0 to 1 do

      for k from 0 to 1 do

         ket||i||j||k:=kr(ket||i||j, ket||k);

      od;

   od;   

od;

## ket||i||j||k||m

for i from 0 to 1 do   

   for j from 0 to 1 do   

      for k from 0 to 1 do   

         for m from 0 to 1 do

            ket||i||j||k||m:=kr(ket||i||j||k, ket||m);

         od;

      od;   

   od;  

od;

## Density operator corresponding to the ket

Ket_To_Rho:=v->evalm(v &* map(conjugate, transpose(v))):

###################################################################

## Procedure:   Make_Ket

## Description: This procedure creates a Ket when given a basis state, 

##              number of qubits and a basis as input. If the value 

##              parameter is too large to be held in a qubit register 

##              of size dim, then the value is taken mod 2^dim.

##              The ket can be created in one of two basis - "std" 

##              or "bell". The "std" basis implies that the value 

##              given is to be converted to the standard basis 

##              element. The "bell" basis implies that the Bell basis 

##              kets are to be produced.

## Inputs:      Int Value
- Decimal representation of the basis state.

##              Int Dim

 - Number of qubits in the ket.

##              Int Basis
- Basis ("std" or "bell")

## Outputs:     Ket K

   - A column vector representing a ket. 

###################################################################

Make_Ket:= proc(value, dim, basis)

  if (basis = 'std') then

    RETURN(Make_Std_Ket(value, dim));

  fi;

  if (basis = 'bell') then

    RETURN(Make_Bell_Ket(value, dim));

  fi;  

  print ("Basis not supported: try std or bell");

  RETURN([]);

end:

###################################################################

## Procedure:   Make_Bell_Ket

## Description: This procedure creates a Bell Basis Ket when given 

##              a basis state, state and a number of qubits  as 

##              input. 

##              If the value parameter is too large to be held in 

##              a qubit register of size dim, then the value is 

##              taken mod 2^dim.

## Inputs:      Int Value
- Decimal representation of the basis state.

##              Int Dim

 - Number of qubits in the ket.

## Outputs:     Ket K

   - A column vector representing a ket. 

###################################################################

Make_Bell_Ket:= proc(value, dim)

   local realval, MSBval, std_1, std_2, BellKet;

   if (dim < 2) then RETURN ([]) fi;

   realval := value mod 2^dim;

   MSBval := iquo(realval, 2^(dim-1) );

   std_1 := Make_Std_Ket(realval, dim);

   std_2 := Make_Std_Ket((2^dim - 1) - realval, dim);

   if (MSBval=0) then

      BellKet:=( std_2 + std_1 );

   else

      BellKet:=( std_2 - std_1 );

   fi;

   BellKet:=( (1/sqrt(2)) * BellKet ); 

   RETURN(evalm(BellKet));

end:  

###################################################################

## Procedure:   Make_Std_Ket

## Description: This procedure creates a Standard basis Ket when  

##              given a basis state, state and a number of qubits  

##              as input.  

##              If the value parameter is too large to be held in 

##              a qubit register of size dim, then the value is 

##              taken mod 2^dim.

## Inputs:      Int Value
- Decimal representation of the basis state.

##              Int Dim

 - Number of qubits in the ket.

## Outputs:     Ket K

   - A column vector representing a ket. 

###################################################################

Make_Std_Ket:= proc(value, dim)

   local answer, realval, realrem, i, k; 

   k||0:=matrix(2,1,[1,0]):

   k||1:=matrix(2,1,[0,1]):

   answer := matrix(1,1,[1]);

   realval := value mod 2^dim; #Reduce input to fit in dim.

   for i from 1 to dim do

     realrem := realval mod 2;     

     if (realrem = 0) then 

       answer := kr(k||0, answer);

     else

       answer := kr(k||1, answer);

     fi;

     realval := iquo(realval,2);

   od;

   RETURN(evalm(answer));

end:  

###################################################################

## Function:    Normalize_Ket

## Description: This procedure normalizes a column vector representing

##              a ket with the 2-norm.

## Inputs:      Ket k
    - Ket to normalize.

## Outputs:     Ket K

   - A column vector representing a 

##                          normalized ket. 

###################################################################

Normalize_Ket:= proc(ket)

  local answer;

  answer := matrix(rowdim(ket),1,normalize(col(ket, 1)));

  RETURN (evalm(answer));

end:

###################################################################

## Function:    RKet

## Description: Convenience function that represent a column vector

##              in symbolic ket form. The ket is represented in the

##              standard basis.

##              Note: This function should not be used in 

##              computations. It should only be used as a "pretty-

##              printer" for readability purposes.

##                                                      

## Inputs:      A column vector representing a ket.

## Outputs:     A column vector in sybolic ket form.

###################################################################

RKet:=proc(ket)

   local r,S,i,logr,bbb;

   r:=rowdim(ket);

   logr:=ceil( evalf(log[2](r)) ); 

   S:=0;

   for i from 1 to r do

      bbb:=Int_To_Bin(i-1,logr);

      S:=S + ket[i,1]*(`|`||bbb||`>`);

   od;

   RETURN(S);

end:

###################################################################

## Procedure:   Probability_Points

## Description: This procedure produces a list of points (2-element 

##              lists) whose x-coordinates are the decimal values 

##              of the basis states in the ket and whose y-coordinates 

##              are the norm of the amplitude squared of that basis state.

## Inputs:      Ket ket     - The input ket.

## Outputs:     List L
     - List of probability points.  

###################################################################

Probability_Points:= proc (ket)

  local i, listpoints;

  listpoints := [];

  for i from 1 to rowdim(ket) do

    listpoints := [ op(listpoints), [i-1, abs(ket[i,1])^2] ];

  od;

end:

###################################################################

## Procedure:   Plot_Ket

## Description: This procedure is equivalent to calling plot 

##              (Probability_Points(ket), style = point). It returns 

##              the maple plot object corresponding to plotting 

##              Probability_Points(ket) with a scatter plot.

## Inputs:      Ket ket     - The ket to plot.

## Outputs:     Plot P
     - Maple plot object.  

###################################################################

Plot_Ket:= proc(ket)

  RETURN(plot(Probability_Points(ket), basis_states = 0..rowdim(ket)-1, probability = 0..1, style=point));

end:

###################################################################

## Procedure:   Measure_One_Qubit

## Description: This procedure returns a 2 element list. The first 

##              element is the ket representing the state of the ket 

##              after the measurement and the second element is the 

##              value measured (0 or 1). 

##              This procedure uses Maple's random number generator 

##              function to perform the measurement.

## Inputs:      Int bit   - An integer < number of qubits in the ket

##              Ket ket
  - Ket to measure.

## Outputs:     List L

  - A list representing the measurement. 

###################################################################

Measure_One_Qubit:= proc(bit, ket)

   local flip, current_bit, counter, i, prob1, prob0, answer, coin,                              winnerprob, erase_bit, result_bit;

   flip := 2^bit;

   if (flip >= rowdim(ket)) then RETURN ([]) fi;

   answer := Normalize_Ket (ket);

   current_bit:=0;

   counter := 0;  

   prob0 := 0;

   prob1 := 0;

   ## This part accumilates probabilities of 0 and 1.

   for i from 1 to rowdim(answer) do

     if (current_bit = 0) then prob0 := prob0 + (abs(answer[i,1]))^2;

     else prob1 := prob1 + (abs(answer[i,1]))^2;

     fi;

     counter := counter + 1;      

     if (counter = flip) then

       counter := 0;

       if (current_bit = 0) then current_bit :=1;

       else current_bit := 0;

       fi;

     fi;

   od;

   ## If the probabilities add to more than 1, something

   ## has gone horribly wrong!

   if (evalf(prob0 + prob1) > 1.1) then RETURN([]); fi;

   ## Flip a coin to see if 0 or 1 wins

   coin := evalf(rand() * 10^(-12));

   if (coin < evalf(prob0) ) then

      erase_bit := false;

      result_bit := 0;   

   else 

      erase_bit := true;

      result_bit := 1;

   fi;

   counter := 0;

   ## Eradicate all the "losing" superposed entries.

   for i from 1 to rowdim(answer) do

      if (erase_bit) then

        answer[i,1] := 0;

      fi;

      counter := counter + 1;

      if (counter = flip) then  

       counter := 0;

       erase_bit := not(erase_bit); 

      fi;

   od;

   ## Re-normalize the answer for the user!

   answer:=Normalize_Ket(answer);

   RETURN([evalm(answer), result_bit]);

end:

###################################################################

## Procedure:   Measure_All_Qubits

## Description: This procedure returns a 2 element list. The first 

##              element is the ket representing the state of the 

##              ket after the measurement. This will be one of the 

##              standard basis states. The second element is an 

##              integer which is a decimal representation of the 

##              basis state measured. 

##              This procedure uses Maple's random number generator 

##              function to perform the measurement.

## Inputs:      Ket ket
  - Ket to measure.

## Outputs:     List L

  - A list representing the measurement. 

###################################################################

Measure_All_Qubits:= proc(ket)

    local answer, coin, answer_value, flag, i, acc;

    answer:=Normalize_Ket(ket);

    coin := evalf(rand() * 10^(-12));

    acc := 0;

    flag := true;

    for i from 1 to rowdim(answer) do

       if flag then acc := evalf(acc + (abs(answer[i,1]))^2); fi;    

       if ((acc >= coin) and flag) then

          flag := false;

          answer [i,1] := 1;

          answer_value := i-1;

       else

          answer [i,1] := 0;

       fi;

    od;

    RETURN ( [evalm(answer),answer_value]);

end:

###################################################################

## Procedure:   General_Measurement

## Description: This procedure performs the measurement of a density 

##              matrix relative to meas_op. 

##              It does not collapse rho into the basis state measured, 

##              however, the function is not random. 

##              It returns a list of 3 element lists. Each element 

##              represents of the possible values measured by this 

##              measurement operator. The first element of each list 

##              is the eigen value of that measured state. The second 

##              integer is the probability of observing that state. 

##              The last entry is a density matrix corresponding to 

##              the state of the system if that state is actually 

##              observed.

## Inputs:      Matrix rho  - Density operator

##              Int meas_op - Measurement operator

## Outputs:     List L

    - A list representing the measurement. 

###################################################################

General_Measurement:= proc (rho, meas_op)

  local eigen,i,j,answer,temp, tempvect, tempmat,non_orth, temp_orth, orth, P, Prob, Result, sum_Pi;

  sum_Pi := matrix(rowdim(rho), rowdim(rho), (i,j)->0);

  answer := [];

  eigen := [eigenvects(meas_op)];

  P := array(1..nops(eigen));

  Prob := array(1..nops(eigen));

  Result := array(1..nops(eigen)); 

  for i from 1 to nops(eigen) do #For each eigenvalue,

    P[i] := matrix(rowdim(rho), rowdim(rho), (i,j)->0);

    temp := convert(op(3,eigen[i]), list);

    non_orth := [];

    for j from 1 to nops(temp) do

      non_orth := [op(non_orth), normalize(temp[j])];

    od;

    temp_orth:=GramSchmidt(non_orth);

    orth := [];

    for j from 1 to nops(temp_orth) do

      orth := [op(orth), convert(transpose(normalize(temp_orth[j])), matrix)];

    od; 

    for j from 1 to nops(orth) do #For each dimention

      tempmat := evalm(orth[j]);

      P[i] :=evalm(matadd(P[i],transpose(tempmat) &* (conjugate(tempmat))));

    od:

    Prob[i] := trace(evalm(P[i] &* rho));

    if (Prob[i] <> 0) then

      Result[i] := evalm(scalarmul(P[i] &* (rho &* P[i]), 1/Prob[i]));

    else 

      Result[i] := evalm(P[i] &* (rho &* P[i]));

    fi; 

    answer:= [op(answer), [eigen[i][1], Prob[i], evalm(map(x->Re(x) + I*Im(x), simplify(Result[i])))] ];

    sum_Pi := evalm(matadd(sum_Pi, P[i]));    

  od:

  RETURN(answer);

end:

###################################################################

## Procedure:   General_Measurement_Ket

## Description: This procedure performs the measurement of a ket 

##              relative to meas_op. 

##              It does not collapse the ket into the basis state 

##              measured, however, the function is not random. 

##              It returns a list of 3 element lists. Each element 

##              represents of the possible values measured by this 

##              measurement operator. The first element of each list 

##              is the eigen value of that measured state. The second 

##              integer is the probability of observing that state. 

##              The last entry is a ket corresponding to the state 

##              of the system if that state is actually observed.

## Inputs:      Ket ket     - Ket to measure

##              Int meas_op - Measurement operator

## Outputs:     List L

    - A list representing the measurement. 

###################################################################

General_Measurement_Ket:= proc(ket, meas_op)  

  local eigen, orth, temp, i, non_orth, num_eigenvects, temp_orth, j, flat_orth, alph, tempmat, Prob, curr_prob, counter, Result, answer;

  num_eigenvects := 0;

  eigen := [eigenvects(meas_op)];

  orth := array(1..nops(eigen));

  flat_orth := [];

  orth:= [];

  Prob := array(1..nops(eigen));

  for i from 1 to nops(eigen) do #For each eigenvalue,

    Prob[i] := 0;

    non_orth := convert(op(3,eigen[i]), list);

    orth := [op(orth), op(2,eigen[i])];

    temp_orth:=GramSchmidt(non_orth);

    for j from 1 to nops(temp_orth) do

      flat_orth:= [op(flat_orth), convert(transpose(normalize(temp_orth[j])),                             matrix)];

    od;    

  od;

  num_eigenvects := nops(flat_orth);

  alph := array(1..num_eigenvects);

  for i from 1 to num_eigenvects do

    tempmat := evalm(flat_orth[i]);

    alph[i] := evalm(conjugate(tempmat) &* (ket))[1,1];

  od;

  curr_prob := 1;

  counter := 0;

  Result := array(1..nops(orth));

  for i from 1 to nops(orth) do

    Result [i] := array(sparse, 1..1,1..rowdim(ket));

  od;

  for i from 1 to num_eigenvects do        

    if (counter = orth[curr_prob]) then

      curr_prob := curr_prob + 1;

      counter := 0;

    fi; 

    counter := counter + 1;

    Prob[curr_prob] := Prob[curr_prob] + eval((abs(alph[i]))^2);

    Result[curr_prob] := matadd( scalarmul(flat_orth[i], (alph[i]) ), Result[curr_prob]); 

  od;   

  for i from 1 to nops(orth) do

    if (Prob[i] <> 0) then

     Result[i] := scalarmul((row(Result[i],1)) , 1/sqrt(Prob[i]));

    fi;

    Result[i] := evalm(simplify(Result[i]));

  od; 

  answer := array(1..nops(orth));

  for i from 1 to nops(orth) do

    answer [i] := [op(1,eigen[i]) ,Prob[i], convert(evalm(map(x->Re(x) + I*Im(x), simplify(Result[i]))), matrix)];

  od;

  RETURN(evalm(answer));

end:

###################################################################

## Procedure:   Small_Transformation

## Description: This function quickly computes the result when we 

##              apply a "small" transformation to a qubit register.

##              This procedure produces a ket which is the result 

##              of applying the input "trans" locally to the 

##              "trans_dim" number of qubits starting with the 

##              "start_bit" bit in the ket.

##

##              NOTE: This function only does the required number of 

##              (nonzero) multiplications as opposed to the 2^(2*dim) 

##              multiplications required when using full-sized 

##              unitary matrices.

## Inputs:      Ket k

       - The register to apply the transformation to.

##              Int dim

     - Number of qubits in the ket (size of 

##                              the register).

##              Int start_bit
- First bit to start the transformation at (<= dim).

##              Int trans_dim
- Number of qubits that trans operates on 

##                              (log base 2 of the size of trans).

##              Trans

       - Small transformation to apply (1-bit 

##                              Hadamard, for example).

## Outputs:     Ket K

       - A transformed ket.  

###################################################################

Small_Transformation:= proc (ket, dim, start_bit, trans_dim, trans)

  local i, j, k, el, left_id_bits, right_id_bits, start_row_position, start_col_position, curr_row_position, curr_col_position,jump_amt, answer, current_multiplier;

  ## Set the answer accumilator to all 0's

  answer := matrix(rowdim(ket), 1, (i,j)->0);

  right_id_bits := 2^(start_bit - trans_dim);

  left_id_bits := 2^(dim - start_bit);

  jump_amt := right_id_bits*rowdim(trans);

  for i from 1 to rowdim (trans) do

   for j from 1 to coldim (trans) do

    current_multiplier := trans[i,j];

    start_row_position:=right_id_bits*(i-1)+1-jump_amt;

    start_col_position:=right_id_bits*(j-1)+1-jump_amt;

    for k from 1 to left_id_bits do

       start_row_position:=start_row_position+jump_amt;

       start_col_position:=start_col_position+jump_amt;

       curr_row_position := start_row_position;

       curr_col_position := start_col_position;

       for el from 1 to right_id_bits do

         answer[curr_row_position, 1]:=answer[curr_row_position,1] + current_multiplier*ket[curr_col_position, 1];

         curr_row_position := curr_row_position + 1;

         curr_col_position := curr_col_position + 1;

       od;

    od;

   od;

  od;

  RETURN (evalm(answer));

end: 

###################################################################

## Function:    Int_To_Bin

## Description: Procedure that converts a positive integer into a 

##              binary number of length lgth.

## Inputs:      n    - A positive integer.

##              lgth - Length of binary number to produce. 

## Outputs:     A binary number.

###################################################################

Int_To_Bin:=proc(n,lgth)

   local nn,s,i,nx;

   nn:=n;   

   s:=``;

   for i from 1 to lgth do

      nx:=irem(nn,2);

      s:=cat(nx,s);

      nn:=iquo(nn,2);

   od;

   RETURN(s);

end:

###################################################################

## Procedure:   Swap_Qubits

## Description: This procedure swaps the first and second bit Hilbert 

##              spaces' orders in the representation of the original 

##              ket. In other words, this functions returns the result 

##              of swapping the positions of qubit bit 1 and qubit 

##              bit 2 in the original ket.

## Inputs:      Ket k

     - The ket to swap.

##

            Int ket_dim
- Number of qubits in the ket (size of the register).

##

            Int bit1

  - First bit to swap.

##

            Int bit2

  - Second bit to swap.

## Outputs:     Ket K

     - Ket returned from swapping.  

###################################################################

Swap_Qubits:= proc (ket, ketdim, bit1, bit2)

  local answer, counter1, counter2, max1, max2, current_bit1, current_bit2, i, temp;

  answer := evalm(ket);

  counter1 := 0;

  counter2 := 0;

  if (bit1 < 1 or bit2 < 1 or bit1 > ketdim or bit2 > ketdim) then

    RETURN([]);

  fi;

  max1 := 2^(bit1-1);

  max2 := 2^(bit2-1);

  current_bit1 := 0;

  current_bit2 := 0;

  for i from 1 to rowdim(ket) do

    if (current_bit1=1 and current_bit2 = 0) then

      temp := answer[i,1];

      answer[i,1] := answer[i+max2-max1, 1];

      answer[i+max2-max1, 1] := temp;      

    fi;

    counter1 := counter1 + 1;

    counter2 := counter2 + 1;

    if (counter1 = max1) then

      counter1 := 0;

      current_bit1 := (-current_bit1) + 1;

    fi;

    if (counter2 = max2) then

      counter2 := 0;

      current_bit2 := (-current_bit2) + 1;

    fi;

  od;

  RETURN(evalm(answer));

end:

###################################################################

## Function:    Bell

## Description: Set of convenience functions that represent the

##              Bell basis kets. The following are supported:

##              Bell||0||0, Bell||0||1, Bell||1||0, Bell||1||1, 

##              Bell||0||0||0, Bell||0||0||1, Bell||0||1||0,

##              Bell||0||1||1, Bell||1||0||0, Bell||1||0||1,

##              Bell||1||1||0, Bell||1||1||1

## Inputs:      None.

## Outputs:     A Bell basis ket.

###################################################################

Bell||0||0:=evalm( 1/sqrt(2) * evalm( ket||0||0 + ket||1||1 ) ):

Bell||0||1:=evalm( 1/sqrt(2) * evalm( ket||0||1 + ket||1||0 ) ):

Bell||1||0:=evalm( 1/sqrt(2) * evalm( ket||0||1 - ket||1||0 ) ):

Bell||1||1:=evalm( 1/sqrt(2) * evalm( ket||0||0 - ket||1||1 ) ):

Bell||0||0||0:=evalm( 1/sqrt(2) * evalm( ket||0||0||0 + ket||1||1||1 ) ):

Bell||0||0||1:=evalm( 1/sqrt(2) * evalm( ket||0||0||1 + ket||1||1||0 ) ):

Bell||0||1||0:=evalm( 1/sqrt(2) * evalm( ket||0||1||0 + ket||1||0||1 ) ):

Bell||0||1||1:=evalm( 1/sqrt(2) * evalm( ket||1||0||0 - ket||0||1||1 ) ):

Bell||1||0||0:=evalm( 1/sqrt(2) * evalm( ket||1||0||0 + ket||0||1||1 ) ):

Bell||1||0||1:=evalm( 1/sqrt(2) * evalm( ket||0||1||0 - ket||1||0||1 ) ):

Bell||1||1||0:=evalm( 1/sqrt(2) * evalm( ket||0||0||1 - ket||1||1||0 ) ):

Bell||1||1||1:=evalm( 1/sqrt(2) * evalm( ket||0||0||0 - ket||1||1||1 ) ):

> Bell4:=proc(a,b,c,d)

   # Bell4(a,b,c,d) = ( ket.(1-a).(1-b).(1-c).(1-d) + ((-1)^a)*ket.a.b.c.d)/sqrt(2)

   # where a,b,c,d are elts of {0,1}

   #

   local  BellKet;

   #

   BellKet:=evalm( ket||(1-a)||(1-b)||(1-c)||(1-d)  );

   if (a=0) then

      BellKet:=evalm( BellKet + ket||a||b||c||d );

   else

      BellKet:=evalm( BellKet - ket||a||b||c||d );

   fi;

   BellKet:=evalm( (1/sqrt(2)) * BellKet );

   RETURN(evalm(BellKet));

end:

for a from 0 to 1 do    for b from 0 to 1 do

   for c from 0 to 1 do    for d from 0 to 1 do

      Bell||a||b||c||d:=Bell4(a,b,c,d);

   od;   od;

      od;   od;

a:='a':b:='b':c:='c':d:='d':

###################################################################

## Function:    Sam

## Description: Set of convenience functions that represent the

##              Sam basis kets. The following are supported:

##              Sam||0||0||0, Sam||0||0||1, Sam||0||1||0,

##              Sam||0||1||1, Sam||1||0||0, Sam||1||0||1,

##              Sam||1||1||0, Sam||1||1||1

## Inputs:      None.

## Outputs:     A Bell basis ket.

###################################################################

Sam||0||0||0:=evalm(1/2 * evalm(ket||1||1||0  + ket||1||0||1 + ket||0||1||1 - ket||0||0||0)):

Sam||0||0||1:=evalm(1/2 * evalm(ket||1||0||0  + ket||0||1||0 - ket||0||0||1 + ket||1||1||1)):

Sam||0||1||0:=evalm(1/2 * evalm(ket||1||0||0  - ket||0||1||0 + ket||0||0||1 + ket||1||1||1)):

Sam||0||1||1:=evalm(1/2 * evalm(ket||1||1||0  + ket||1||0||1 - ket||0||1||1 + ket||0||0||0)):

Sam||1||0||0:=evalm(1/2 * evalm(-ket||1||0||0 + ket||0||1||0 + ket||0||0||1 + ket||1||1||1)):

Sam||1||0||1:=evalm(1/2 * evalm( ket||1||1||0 - ket||1||0||1 + ket||0||1||1 + ket||0||0||0)):

Sam||1||1||0:=evalm(1/2 * evalm(-ket||1||1||0 + ket||1||0||1 + ket||0||1||1 + ket||0||0||0)):

Sam||1||1||1:=evalm(1/2 * evalm( ket||1||0||0 + ket||0||1||0 + ket||0||0||1 - ket||1||1||1)):

###################################################################

## Procedure:   kr

## Description: Computes the Kronecker product (tensor product) of 

##              two matrices.

## Inputs:      A - An n x m matrix

##              B - A  p x q matrix

## Outputs:     C - An np x mq matrix

###################################################################

kr:=proc(A,B)

   local mm, nn, ans, ans1;

   ans:=[];

   for mm from 1 to rowdim(A) do

      for nn from 1 to coldim(A) do

         ans:=[op(ans), scalarmul(B, A[mm,nn] )];

      od;

   od;

   ans1:=blockmatrix(rowdim(A), coldim(A), ans );

   RETURN(evalm(ans1));

end:

###################################################################

## Procedure:   QF

## Description: Computes a unitary matrix Q whose rows are the unit

##              eigenvectors of a unitary matrix  U.

## Inputs:      U - A unitary matrix

## Outputs:     Q - A Diagonalized unitary matrix

###################################################################

QF:=proc(U)

   local  Dim_U,List,n,nu,Q,j,temp_j,n_j,List_j,List_jj,i,Vec_i,Norm_Vec_i,k;

   Dim_U:=rowdim(U):

   List:=[eigenvects(U)];   

   n:=nops(List);

   nu:=1;

   Q:=matrix(Dim_U,Dim_U);

   for j from 1 to n do

      temp_j:=op(j,List); 

      n_j:=op(2,temp_j); 

      List_j:= convert(op(3,temp_j), list);

      List_jj:=GramSchmidt(List_j);

      for i from 1 to n_j do

         Vec_i:=convert( op(i, List_jj), vector );

            Norm_Vec_i:=norm(eval(Vec_i),2);

         for k from 1 to Dim_U do  

            Q[nu,k]:=Vec_i[k]/Norm_Vec_i; 

         od; 

         nu:=nu+1;

      od;

   od; 

   Q:=map(radnormal, Q); 

   RETURN (evalm(Q));

end:

###################################################################

## Function:    L3x3, LL

## Description: Set of convenience functions that represent 

#               predefined infinitesimal generators.

## Inputs:      None.

## Outputs:     Matrix representing SO(3) or SO(4)

###################################################################

## L3x3.j ... j=1..3 ... are predefined infinitesimal generators of SO(3)

L3x3||0:=matrix(3,3, [1,0, 0,   0,1,0,  0, 0,1]):

L3x3||1:=matrix(3,3, [0,0, 0,   0,0,1,  0,-1,0]):

L3x3||2:=matrix(3,3, [0,0,-1,   0,0,0,  1, 0,0]):

L3x3||3:=matrix(3,3, [0,1, 0,  -1,0,0,  0, 0,0]):

for i from 0 to 3 do

   LLL||i:=evalm(L3x3||i);

od:

for i from 0 to 3 do   

   for j from 0 to 3 do

      LLL||i||j:=kr(LLL||i, LLL||j);

   od;   

od;

## LL.j ... j=1..6 ... are predefined infinitesimal generators of SO(4)

LL||1:=matrix(4,4,[0,1,0,0, -1,0,0,0,   0, 0,0,0,   0, 0, 0,0]):

LL||2:=matrix(4,4,[0,0,1,0,  0,0,0,0,  -1, 0,0,0,   0, 0, 0,0]):

LL||3:=matrix(4,4,[0,0,0,1,  0,0,0,0,   0, 0,0,0,  -1, 0, 0,0]):

LL||4:=matrix(4,4,[0,0,0,0,  0,0,1,0,   0,-1,0,0,   0, 0, 0,0]):

LL||5:=matrix(4,4,[0,0,0,0,  0,0,0,1,   0, 0,0,0,   0,-1, 0,0]):

LL||6:=matrix(4,4,[0,0,0,0,  0,0,0,0,   0, 0,0,1,   0, 0,-1,0]):

###################################################################

## Function:    FLL4

## Description: This procedure expresses a 4x4 skew-smmetric matrix 

#               as the sum of a linear combination of LL.i's.

## Inputs:      LU = Sum( i=1..6, A[1,i]*LL.i ).

## Outputs:     Matrix representing the sum of a linear combination 

##              of LL.i's.

###################################################################

FLL4:=proc(LU)

   local  A, i;

   A:=matrix(1,6);

   for i from 1 to 6 do

      A[1,i]:=trace( evalm( LL||i &* LU) )/(-2);

   od;

   RETURN(evalm(A));

end:

> for i from 0 to 3 do   

   for j from 0 to 3 do

      SS||(i+4*j):=kr(sigma||i, sigma||j)

   od; 

od;

for i from 0 to 3 do

   SSS||i:=evalm(sigma||i);

od:

for i from 0 to 3 do   

   for j from 0 to 3 do

      SSS||i||j:=evalm(SS||(i+4*j));

   od;   

od;

for i from 0 to 3 do   

   for j from 0 to 3 do 

      for k from 0 to 3 do

         SSS||i||j||k:=kr(SS||(i+4*j), sigma||k):

      od;

   od;   

od;

for i from 0 to 3 do   

   for j from 0 to 3 do

      ij:=i+4*j; 

      for k from 0 to 3 do  

         for m from 0 to 3 do

            SSS||i||j||k||m:=kr(SS||(ij), SS||(k+4*m)):

         od;   

      od;

   od;   

od;

###################################################################

## Procedure:   Logz

## Description: Computes the natural log of integer z.

## Inputs:      

## Outputs:     

###################################################################

Logz:=proc(z)

   local  zpolar;   

   if (z=0) then

      RETURN(0);

   else

      zpolar:=convert(z,polar);

      RETURN( (I*op(2,zpolar)) + ln(op(1,zpolar))   );

   fi;

end:  

###################################################################

## Procedure:   LogU

## Description: Computes the natural log of a unitary matrix U.

## Inputs:      Unitary matrix U.

## Outputs:     Matrix M which is the natural log of matrix U.

###################################################################

LogU:=proc(U)

   local  Q, IQ, DU, LDU, LU;

   Q   := QF(U);

   IQ  := map(conjugate,transpose(Q));

   DU  := map(radnormal,evalm( Q &* U &* IQ));

   LDU := map(Logz, DU);

   LU  := map(radnormal, evalm( IQ &* LDU &* Q));

   RETURN(evalm(LU));

end:

###################################################################

## Procedure:   Singer2Log

## Description: Computes multiple valued log of matrix W.

## Inputs:      Matrix W.

## Outputs:     Matrix M which is the multi-valued log of matrix W.

###################################################################

Singer2Log:=proc(W)

   local LW,UQ,IUQ,DD,n1,n2,n3,n4;

   LW:=LogU(W):

   UQ:=QF(W):

   IUQ:=map(conjugate,transpose(UQ)):

   DD:=map(radnormal, evalm( IUQ &* diag(n1,n2,n3,n4) &* UQ)):

   RETURN(evalm(LW + DD));

end:

###################################################################

## Procedure:   Pauli1

## Description: This procedure expresses a 2x2 skew-Hermitian matrix

##              as the sum of Pauli matrices. 

## Inputs:      LU:= Sum(i=0..3, R[i+1]*sigma.i)

##              Note that the index of the matrix R runs from 1 to 4.

## Outputs:     Sum of the Pauli matrices.

###################################################################

Pauli1:=proc(LU)   

   local  A, i;

   A:=matrix(1,4);

   for i from 0 to 3 do

      A[1,i+1]:=trace( evalm( sigma||i &* LU) )/2;

   od;

   RETURN(evalm(A));

end:

###################################################################

## Procedure:   RPauli1

## Description: Expresses density operator rho as a formal sum of 

##              sigma.i's. 

## Inputs:      Density operator rho.

## Outputs:     Density operator as formal sum of sigma.i's.

###################################################################

RPauli1:=proc(rho)

   local R,RSUM,i;

   R:=Pauli1(rho);

   RSUM:=0;

   for i from 0 to 3 do    

      RSUM:=RSUM + R[1,i+1]*S||i;

   od;   

   RETURN(RSUM);

end:

###################################################################

## Procedure:   Pauli2

## Description: This procedure expresses a 4x4 skew-Hermitian matrix

##              as the sum of the tensor product of Pauli matrices. 

## Inputs:      LU:= Sum( i=0..3, j=0..3, R[i+1,j+1]*SS.(i+4*j) )

##              Note that the index of the matrix R runs from 1 to 4.

## Outputs:     Sum of the tensor product of Pauli matrices.

###################################################################

Pauli2:=proc(LU)

   local  A, i, j;

   A:=matrix(rowdim(LU),coldim(LU));

   for i from 0 to 3 do

      for j from 0 to 3 do

         A[i+1,j+1]:=trace( evalm( SS||(i+4*j) &* LU) )/4;

      od;

   od;

   RETURN(evalm(A));

end:

###################################################################

## Procedure:   RPauli2

## Description: Expresses density operator rho as a formal sum of 

##              sigma.i(X)sigma.j's. 

## Inputs:      Density operator rho.

## Outputs:     Density operator as formal sum of sigma.i(X)sigma.j's.

###################################################################

RPauli2:=proc(rho)

   local R,RSUM,i,j;

   R:=Pauli2(rho);

   RSUM:=0;

   for i from 0 to 3 do   for j from 0 to 3 do 

      RSUM:=RSUM + R[i+1,j+1]*S||i||j;

   od;   od;

   RETURN(RSUM);

end:

###################################################################

## Procedure:   Inv_Pauli2

## Description: Given a 4x4 matrix R over the reals, this procedure 

##              computes  rho = Sum(p=0..3, q=0..3) R[p+1,q+q]*sigma.p(X)sigma.q. 

## Inputs:      R - 4x4 matrix over the reals.

## Outputs:     Sum(p=0..3, q=0..3) R[p+1,q+q]*sigma.p(X)sigma.q.

###################################################################

Inv_Pauli2:=proc(R)

   local p,q,rho;

   rho:=diag(0,0,0,0):

   for p from 0 to 3 do   

      for q from 0 to 3 do

         rho:= evalm( rho + evalm(R[p+1,q+1]*SS||(p+4*q)) );

      od;   

   od;

   RETURN(evalm(rho));

end:

###################################################################

## Procedure:   Pauli3

## Description: This procedure expresses a 8x8 skew-Hermitian matrix

##              as the sum of the tensor product of Pauli matrices. 

## Inputs:      LU:= Sum( i=0..3, j=0..3, k=0..3, R[i,j,k]*SSS.i.j.k )

##              Note that the index of the matrix R runs from 0 to 3.

## Outputs:     Sum of the tensor product of Pauli matrices.

###################################################################

Pauli3:=proc(LU)

   local  A, i, j, k;

   A:=array(0..3,0..3,0..3);

   for i from 0 to 3 do

      for j from 0 to 3 do

         for k from 0 to 3 do

            A[i,j,k]:=trace( evalm(SSS||i||j||k &* LU) )/8;

         od;

      od;

   od;

   RETURN(eval(A));

end:

###################################################################

## Procedure:   RPauli3

## Description: Expresses density operator rho as a formal sum of 

##              sigma.i(X)sigma.j(X)sigma.k's. 

## Inputs:      Density operator rho.

## Outputs:     Density operator as formal sum of sigma.i(X)sigma.j(X)sigma.k's.

###################################################################

RPauli3:=proc(rho)

   local R,RSUM,i,j,k;

R:=Pauli3(rho);

   RSUM:=0;

   for i from 0 to 3 do   

      for j from 0 to 3 do   

         for k from 0 to 3 do

            RSUM:=RSUM + R[i,j,k]*S||i||j||k;

         od;   

      od;   

   od;

   RETURN(RSUM);

end:

###################################################################

## Procedure:   Pauli4

## Description: This procedure expresses a 16x16 skew-Hermitian matrix

##              as the sum of the tensor product of Pauli matrices. 

## Inputs:      LU:= Sum( i=0..3, j=0..3, k=0..3, m=0..3, R[i,j,k,m]*SSS.i.j.k.m )

##              Note that the index of the matrix R runs from 0 to 3.

## Outputs:     Sum of the tensor product of Pauli matrices.

###################################################################

Pauli4:=proc(LU)

   local  A, i, j, k, m;

   A:=array(0..3,0..3,0..3,0..3);

   for i from 0 to 3 do   

      for j from 0 to 3 do

         for k from 0 to 3 do   

            for m from 0 to 3 do

               A[i,j,k,m]:=trace( evalm( SSS||i||j||k||m &* LU) )/16;

            od;   

         od;

      od;   

   od;

   RETURN(eval(A));

end:

###################################################################

## Procedure:   RPauli4

## Description: Expresses density operator rho as a formal sum of 

##              sigma.i(X)sigma.j(X)sigma.k(X)sigma.m's. 

## Inputs:      Density operator rho.

## Outputs:     Density operator as formal sum of 

##              sigma.i(X)sigma.j(X)sigma.k'(X)sigma.m's.

###################################################################

RPauli4:=proc(rho)

   local R,RSUM,i,j,k,m;

   R:=Pauli4(rho);

   RSUM:=0;

   for i from 0 to 3 do   

      for j from 0 to 3 do   

         for k from 0 to 3 do   

            for m from 0 to 3 do

               RSUM:=RSUM + R[i,j,k,m]*S||i||j||k||m;

            od;   

         od;   

      od;   

   od;

   RETURN(RSUM);

end:

###################################################################

## Procedure:   Matrep

## Description: This procedure converts a permutation represented 

##              as a product of (not necessarily disjoint) cycles 

##              to a permutation matrix.

##              It takes two arguments, Perm and P_Deg. Perm is 

##              expected to be a list of lists of integers and P_Deg 

##              is expected to be an integer. Perm is interpreted 

##              as a permutation written as a product of (not 

##              necessarily disjoint) cycles. Matrep will return 

##              a square permutation matrix U of dimension P_Deg by 

##              P_Deg corresponding to the permutations in Perm.

## Inputs:      Perm
   - a list of lists of integers.

##              P_Deg   - an integer.

## Outputs:     Matrix M - square permutation matrix U of dimension 

##              P_Deg by P_Deg corresponding to the permutations in 

##              Perm.

###################################################################

Matrep:=proc(P, P_Deg)

   local  f, Pmat, n, j, Cycle_j, n_j, a_j, i, b_j;

   f:=(ii,jj)->if(ii=jj) then 1 else 0 fi;

   Pmat:=matrix(P_Deg, P_Deg, f);

   n:=nops(P);

   if (n<1) then 

      RETURN(evalm(Pmat));

   else

      for j from 1 to n do

         Cycle_j:=op(j,P);

         n_j:=nops(Cycle_j);

         if (n_j>1) then

            a_j:=op(1,Cycle_j);

            for i from 2 to n_j do

               b_j:=op(i,Cycle_j);

               Pmat:=swaprow(Pmat,a_j+1,b_j+1);

            od;

         fi;

      od;

   fi;

   RETURN(evalm(Pmat));

end:

> RemoveIPi:=proc(A)

   local  ff, M;  

   ff:=(i,j)-> if (i=j) then 1/(Pi*I) else  0  fi;

   M:=evalm( matrix(rowdim(A),coldim(A),ff) &* A);

   RETURN(evalm(M));

end:

###################################################################

## Functions:   f0, f1, f2, f3, f4, f5

## Description: Helper functions for the patial trace functions. 

## Inputs:      None.

## Outputs:     None.     

###################################################################

f||0:=x->2*x:

f||1:=x->irem(x,2  ) + ( iquo(x,2  )*2^2):

f||2:=x->irem(x,2^2) + ( iquo(x,2^2)*2^3):

f||3:=x->irem(x,2^3) + ( iquo(x,2^3)*2^4):

f||4:=x->irem(x,2^4) + ( iquo(x,2^4)*2^5):

f||5:=x->irem(x,2^5) + ( iquo(x,2^5)*2^6):

###################################################################

## Procedure:   Trace1

## Description: Partial trace function 1. 

## Inputs:      A - square matrix

## Outputs:     Trace value of Matrix A.     

###################################################################

Trace1:=proc(A)

   local ans, ii, jj;   

   ans:=matrix(rowdim(A)/2, coldim(A)/2);

   for ii from 1 to rowdim(A)/2 do;

      for jj from 1 to coldim(A)/2 do;

         ans[ii,jj]:=A[1+f||0(ii-1),1+f||0(jj-1)] + A[1+f||0(ii-1)+1,1+f||0(jj-1)+1];  

      od;

   od;

   RETURN(evalm(ans))

end:

Trace2:=proc(A)

   local ans, ii, jj;   

   ans:=matrix(rowdim(A)/2, coldim(A)/2  );

   for ii from 1 to rowdim(A)/2 do;

      for jj from 1 to coldim(A)/2 do;

         ans[ii,jj]:=A[1+f||1(ii-1),1+f||1(jj-1)] + A[1+f||1(ii-1)+2,1+f||1(jj-1)+2];  

      od;

   od;

   RETURN(evalm(ans))

end:

Trace3:=proc(A)

   local ans, ii, jj;   

   ans:=matrix(rowdim(A)/2, coldim(A)/2  );

   for ii from 1 to rowdim(A)/2 do;

      for jj from 1 to coldim(A)/2 do;

         ans[ii,jj]:=A[1+f||2(ii-1),1+f||2(jj-1)] + A[1+f||2(ii-1)+4,1+f||2(jj-1)+4];  

      od;

   od;

   RETURN(evalm(ans))

end:

Trace4:=proc(A)

   local ans, ii, jj;   

   ans:=matrix(rowdim(A)/2, coldim(A)/2  );

   for ii from 1 to rowdim(A)/2 do;

      for jj from 1 to coldim(A)/2 do;

         ans[ii,jj]:=A[1+f||3(ii-1),1+f||3(jj-1)] + A[1+f||3(ii-1)+8,1+f||3(jj-1)+8];  

      od;

   od;

   RETURN(evalm(ans))

end:

Trace5:=proc(A)

   local ans, ii, jj;   

   ans:=matrix(rowdim(A)/2, coldim(A)/2  );

   for ii from 1 to rowdim(A)/2 do;

      for jj from 1 to coldim(A)/2 do;

         ans[ii,jj]:=A[1+f||4(ii-1),1+f||4(jj-1)] + A[1+f||4(ii-1)+16,1+f||4(jj-1)+16];  

      od;

   od;

   RETURN(evalm(ans))

end:

Trace6:=proc(A)

   local ans, ii, jj;   

   ans:=matrix(rowdim(A)/2, coldim(A)/2  );

   for ii from 1 to rowdim(A)/2 do;

      for jj from 1 to coldim(A)/2 do;

         ans[ii,jj]:=A[1+f||5(ii-1),1+f||5(jj-1)] + A[1+f||5(ii-1)+32,1+f||5(jj-1)+32];  

      od;

   od;

   RETURN(evalm(ans))

end:

# Partial trace of the n-th qubit = (Trace.(n+1))(rho)  

# Qubits are listed right to left, and labeled 0,1, ...

# n = 0,1,2,3,4,5

PTrace:=(n,rho)->(Trace||(n+1))(rho):

###################################################################

## Procedure:   TraceRL, Trace LR

## Description: B[i1,j0] = Sum(a=0..1, A[a+2*i1,j0+2*a] ). 

## Inputs:      A - square matrix

## Outputs:     Trace value of Matrix A.     

###################################################################

TraceRL:=proc(A)

   local B;

   B:=matrix(2,2);

   B[1,1]:=A[1,1]+A[2,3];   B[1,2]:=A[1,2]+A[2,4];

   B[2,1]:=A[3,1]+A[4,3];   B[2,2]:=A[3,2]+A[4,4];

   RETURN(evalm(B));

end:

TraceLR:=proc(A)

   local B;

   # B[i0,j1] = Sum(a=0..1, A[i0+2*a,a+2*j1] )

   B:=matrix(2,2);

   B[1,1]:=A[1,1]+A[3,2];   B[1,2]:=A[1,3]+A[3,4];

   B[2,1]:=A[2,1]+A[4,2];   B[2,2]:=A[2,3]+A[4,4];

   RETURN(evalm(B));

end:

###################################################################

## Procedure:   Fourier

## Description: This procedure computes the nxn matrix (omega^ij)/sqrt(n), 

##              where omega is the primitive n-th root of unity 

##              given by omega=exp(2PiI/n).

## Inputs:      n - An integer

## Outputs:     nxn Matrix M.     

###################################################################

Fourier:=proc(n)

   local omega,A,i,j;

   omega:=exp(2*Pi*I/n);

   A:=matrix(n,n);

   for i from 0 to n-1 do   for j from 0 to n-1 do

      A[i+1,j+1]:=evalc(omega^(i*j));

   od;   od;

   A:=evalm( 1/sqrt(n) * A);

   RETURN(evalm(A));

end:

###################################################################

## Procedure:   Vec

## Description: This procedure transforms an mxn matrix A into an 

##              mxn column vector V, column by column. 

## Inputs:      A - An mxn matrix 

## Outputs:     V - An mxn column vector     

###################################################################

Vec:=proc(A)

   local m,n,V_Dim,V,i,j;

   # Vec transforms an mxn matrix A into an m*n column vector V, column by column   

   m:=rowdim(A); n:=coldim(A); V_Dim:=m*n;

   V:=vector(V_Dim);

   for j from 0 to n-1 do   for i from 0 to m-1 do

      V[i+m*j +1]:=A[i+1,j+1];

   od;   od;

   RETURN(evalm(V));

end:

###################################################################

## Procedure:   Ad

## Description: This procedure computes Ad_Q as a 16x16 matrix. 

## Inputs:      Q - A 4x4 unitary matrix 

## Outputs:     M - A 16x16 matrix     

###################################################################

Ad:=proc(Q)

   local  M,p,q,Apq,R_pq,Vpq,i;

   M:=matrix(16,16); 

   for q from 0 to 3 do   print([`q=`||q]); for p from 0 to 3 do       

      #print([p,q]);

      Apq:=evalm( Q &* SS||(p+4*q) &* My_Adjoint(Q) );

      R_pq:=Pauli2(Apq);

      Vpq:=Vec(R_pq);

      for i from 0 to 15 do

         #print([i]);

         M[i+1,p+4*q +1]:=Vpq[i+1];

         #print([i]);

      od;

   od; od;

   RETURN(evalm(M));

end:

###################################################################

## Procedure:   Small_ad

## Description: This procedure computes small ad_Q as a 16x16 matrix. 

##              Small_ad(Q) (M) = ad_iQ(M), where M is assumed to be 

##              skew Hermitian.

## Inputs:      Q - A 4x4 skew Hermitian matrix 

## Outputs:     M - A 16x16 matrix     

###################################################################

Small_ad:=proc(Q)

   local  M,p,q,Apq,R_pq,Vpq,i;

   M:=matrix(16,16); 

   for q from 0 to 3 do   print([`q=`||q]); for p from 0 to 3 do

      Apq:=evalm( evalm( evalm(Q &* SS||(p+4*q)) - evalm(SS||(p+4*q) &* Q)) );

      Apq:=evalm( I * Apq);

      R_pq:=evalm(-I*Pauli2(Apq));

      Vpq:=Vec(R_pq);

      for i from 0 to 15 do

         M[i+1,p+4*q +1]:=Vpq[i+1];

      od;

   od; od;

   RETURN(evalm(M));

end:
###################################################################

## Procedure:   FCom

## Description: This procedure takes two square matrices of the   

##              same size and computes the commutator of A and B 

##              defined by AB - BA. 

## Inputs:      A - An nxn square matrix 

##              B - An nxn square matrix

## Outputs:     M - The commutator [A,B] of matrices A and B     

###################################################################

FCom:=proc(A,B)

   local X;

   X:=evalm( evalm(A &* B) - evalm(B &* A) );

   RETURN(evalm(X));

end:

## Computes the conjugate of complex number z.

My_Conj:=x->eval(subs( I=-I,x)):

## Computes the adjoint of the complex square matrix A

My_Adjoint:=A->map( My_Conj, transpose(A)):

Ket:=(a,b)->matrix(2,1,[cos(a),expand((exp(I*b)*sin(a)))]):

###################################################################

## Procedure:   Id

## Description: This procedure takes a positive integer n, and   

##              returns the n x n identity matrix. 

## Inputs:      n - A positive integer 

## Outputs:     I - The nxn Identity matrix     

###################################################################

Id:=proc(n)

   local  f,i,j, A;   

   f:=(i,j)->if (i=j) then 1 else 0 fi:

   A:=matrix(n,n,f);

   RETURN(evalm(A));

end:

###################################################################

## Procedure:   CNOT

## Description: Permutation (written as a product of disjoint cyles) 

##              which represents a CNOT wit target bit T, control 

##              bit C, and with W wires labeled 0..(W-1).

## Convention:  First=Bottom =Right and Last=Top=Left 

## Inputs:      T - Target bit

##              C - Control bit

##              W - Wires labeled 0..(W-1) 

## Outputs:     Permutation written as a product of disjoint cycles     

###################################################################

CNOT:=proc(T,C,W)

   local  ET,EC,SETEC,EW,L,k,BITC,BITT; 

   ET:=2^T;   EC:=2^C;   SETEC:=ET+EC; EW:=2^W;

   L:=[];

   for k from 0 to (EW-1) do

      BITC:=irem(floor(k/EC),2);   

      BITT:=irem(floor(k/ET),2);

      if (BITC+BITT=0) then

         L:=[op(L), [EC+k, SETEC+k] ];

      fi;

   od;

   RETURN(L);

end:

CNOT12:=matrix(4,4,[1,0,0,0,   0,1,0,0,   0,0,0,1,   0,0,1,0]):

CNOT21:=matrix(4,4,[1,0,0,0,   0,0,0,1,   0,0,1,0,   0,1,0,0]):

###################################################################

## Procedure:   Hadamard

## Description: Hadamard(n) = (X)_(i=1..n) H, where H is the 2x2 

##              Hadarmard matrix.

## Inputs:      n - A positive integer 

## Outputs:     H - 2x2 Hadamard matrix     

###################################################################

Hadamard:=proc(n)

   local IR2,H,Ans,i;

   IR2:=1/sqrt(2);

   H:=matrix(2,2,[IR2,IR2,IR2,-IR2]);

   Ans:=evalm(H);

   if (n<>1) then

      for i from 1 to n-1 do

         Ans:=evalm( kr(Ans,H));

      od;

   fi;

   RETURN(evalm(Ans));

end:

###################################################################

## Procedure:   Entropy

## Description: Entropy of a Hermitian operator.

##              This procedure has not been completed.

##              The problem is that the eigenvalues, although

##              real, are not always non-negative.  A second

##              problem is how to normalize the Hermitian matrix

##              before computing the entropy.  Should one make it

##              of 2-norm one?  Should one make it of trace one?

## Inputs:      A - Hermitian operator 

## Outputs:     Entropy of the Hermition operator     

###################################################################

Entropy:=proc(A)

   local EV,nEV,Ans,n;

   EV:=vector([eigenvals(evalm((-I)*A))]);

   nEV:=1/norm(EV,2);

   EV:=evalm( nEV * EV);

   print(`EV=`,EV);

   Ans:=0;

   for n from 1 to nops(EV) do

      Ans:=Ans - EV[n]*Logz(EV[n]);

   od;

   RETURN(Ans):

end:

###################################################################

## Procedure:   norm2

## Description: Computes the square root of the spectral radius of 

##              evalm(X &* map(conjugate, transpose(X))).

## Inputs:      n - A positive integer 

## Outputs:     H - 2x2 Hadamard matrix     

###################################################################

norm2:=X-> sqrt(max(op(map(abs,[eigenvals(

      evalm( X &* map(conjugate,transpose(X)) ))])))):

